nCells 4-param double logistic.nb

This *Mathematica* 7 notebook presents, generates data from and retrieves the parameters of a 4-parameter continuous deterministic model of microbial growth and mortality, nCells4[t]. It is constructed from an underlying 3-parameter logistic model of cell's division probability, pDiv4[t], and 3-parameter logistic model of cell's mortality, pMort4[t] having the same asymptote and steepness parameters, pAsym and k, but different characteristic times, tcd and tcm, respectively. As a test, growth/morality data are first generated with this model and then the NonlinearModelFit function is used to fit them and retrieve the values of the 4 original generation parameters. The generated data and the fitted model are then plotted together for visual comparison. Also plotted are the corresponding probability functions pDiv4[t] and pMort4[t]. This is followed by the generation with the same model of some noisy data having a random scatter. The NonlinearModelFit function is used again in an attempt to recover the model's 4 parameters. The generated noisy data and the fitted curve are then plotted together for visual comparison and so are the new probability functions pDiv4[t] and pMort4[t]. The generated noisy data can be replaced by pasted experimental data and the procedure used to estimate the organism's growth/mortality parameters and the corresponding probability functions pDiv4[t] and pMort4[t] under the pertinent conditions.

Programmed by Mark D. Normand based on models developed and elaborated by Micha Peleg, Joseph Horowitz, Murray Eisenberg and Maria G. Corradini. Last modified: November 11, 2009

Clear all variables used in this notebook.

In[445]:=
```
Clear[ dt, maxNoise, n0, nCells4, nCells4Pts, nCells4PtsRand, nlr4Results, nlr4ResultsRand,
  nParams, nPts4, nPts4Rand, pDiv4, pMort4, pAsym, k, tcd, tcm, tSt, tEn, plot4pDivpMortGen,
  plot4pDivpMortExact, plot4pDivpMortRand, plot4GenExact, plot4FitExact, plot4FitGenExact,
  plot4GenRand, plot4FitRand, plot4FitGenRand, seed, yDivMortMax, yMax, yMaxRand]
```

The pDiv4[t] function defines the probability of cell division at time t. The parameters of pDiv4[t] are pAsym, k and tcd.

In[446]:=
```
pDiv4[t_] := pAsym / (1. + Exp[k * (tcd - t)])
```

The pMort4[t] function defines the probability of cell mortality at time t. The parameters of pMort4[t] are pAsym, k and tcm.

In[447]:=
```
pMort4[t_] := pAsym / (1. + Exp[k * (tcm - t)])
```

n0 is the initial number of cells alive at time t=0.

In[448]:=
```
n0 = 1000.;
```

Define the nCells4Integrate[t] function that combines the pDiv4[t] and pMort4[t] functions to give the number of cells alive at time t.

In[449]:=
```
nCells4Integrate[t_] :=
  n0 * Exp[Subtract @@ (Integrate[pDiv4[tt] - pMort4[tt], tt] /. tt → {t, 0})]
```

If both pDiv4[t] and pMort4[t] can be integrated anlytically, calling nCells4Integrate[t] gives an explicit formula that does NOT include Integrate.

In[450]:= **nCells4Integrate[t]**

Out[450]= $1000.\, e^{-pAsym\left(\frac{\text{Log}\left[1.+e^{k\,tcd}\right]}{k}-\frac{1.\,\text{Log}\left[1.+e^{k\,tcm}\right]}{k}\right)+pAsym\left(0.\,t+\frac{\text{Log}\left[1.+e^{-1.\,k\,t+k\,tcd}\right]}{k}-\frac{1.\,\text{Log}\left[1.+e^{-1.\,k\,t+k\,tcm}\right]}{k}\right)}$

Now define the nCells4[t] function to be the explicit formula shown above WHICH DOES NOT include Integrate.

In[451]:= **nCells4[t_] = nCells4Integrate[t]**

Out[451]= $1000.\, e^{-pAsym\left(\frac{\text{Log}\left[1.+e^{k\,tcd}\right]}{k}-\frac{1.\,\text{Log}\left[1.+e^{k\,tcm}\right]}{k}\right)+pAsym\left(0.\,t+\frac{\text{Log}\left[1.+e^{-1.\,k\,t+k\,tcd}\right]}{k}-\frac{1.\,\text{Log}\left[1.+e^{-1.\,k\,t+k\,tcm}\right]}{k}\right)}$

nParams is the number of different parameters in the nCells4[t] model. Assign initial values to all 4 parameter variables. These values will be used to generate a list of ordered-pair {t, nCells4[t]} data values.

In[452]:= **nParams = 4; pAsym = 0.4; k = 0.2; tcd = 0.; tcm = 8.;**

tSt and tEn are the starting and ending times over which to generate and plot the data. dt is the delta-t time increment between data points.

In[453]:= **tSt = 0.; tEn = 40.; dt = 4.;**

minOffset is used in the plots to force plotting the axis tick mark label at 0.

In[454]:= **minOffset = 0.001;**

Assign the maximum y-axis limit of the pDiv4 and pMort4 plots to yDivMortMax.

In[455]:= **pDiv4[tEn]**

Out[455]= 0.399866

In[456]:= **pMort4[tEn]**

Out[456]= 0.399336

In[457]:= **yDivMortMax = 0.5;**

Plot the pDiv4[t] and pMort4[t] functions together on the same plot using the currently assigned values of the 4 parameters used to generate the exact data.

In[458]:=
```
plot4pDivpMortGen = Plot[{pDiv4[t], pMort4[t]}, {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True,
  PlotLabel → ToString[nParams] <> "-par. data generation pDiv4[t] (blue) & pMort4[t] (red)",
  FrameLabel → {"time", "pDiv4[t] & pMort4[t]"}]
```

Out[458]=



Generate, fit and plot some exact data (nCells4Pts) using the 4-parameter nCells4[t] function.

nCells4Pts is a generated list of {t, nCells4[t]} ordered data pairs to be used in curve fitting with nonlinear regression and plotting.

In[459]:=
```
nCells4Pts = Table[{t, nCells4[t]}, {t, tSt, tEn, dt}]
```

Out[459]=
```
{{0., 1000.}, {4., 1788.73}, {8., 3199.56}, {12., 5017.73}, {16., 6643.32}, {20., 7722.47},
 {24., 8314.43}, {28., 8607.12}, {32., 8744.59}, {36., 8807.62}, {40., 8836.21}}
```

nPts4 is the number of generated data points in the nCells4Pts list.

In[460]:=
```
nPts4 = Length[nCells4Pts]
```

Out[460]=
```
11
```

Assign the maximum y-axis value to be used in the exact data plots to yMax.

In[461]:=
```
Max[nCells4Pts[[All, 2]]]
```

Out[461]=
```
8836.21
```

In[462]:=
```
yMax = 9000.;
```

Clear all the variables to be used by NonlinearModelFit.

In[463]:=
```
Clear[nlr4Results, pAsym, k, tcd, tcm]
```

Call the NonlinearModelFit function which adjusts the values of the 4 parameters to give the best fit of the nCells4[t] function to the nCells4Pts data and return a report in nlr4Results. All 4 parameters are unconstrained.

In[464]:=
```
nlr4Results = NonlinearModelFit[nCells4Pts, nCells4[t],
  {{pAsym, 0.5}, {k, 0.5}, {tcd, 1.}, {tcm, 10.}}, t, MaxIterations → 500]
```

NonlinearModelFit::sszero :
The step size in the search has become less than the tolerance prescribed by the PrecisionGoal option, but the gradient is larger than the tolerance specified by the AccuracyGoal option. There is a possibility that the method has stalled at a point that is not a local minimum. ≫

Out[464]= FittedModel[$1000.\,e^{2.18151\ll 1\gg \ll 19\gg \ll 1\gg \ll 1\gg}$]

Show the best fitted model, the best fit parameters, the parameter table, the R-squared coefficient of determination, the value of R-squared adjusted for the number of model parameters, the Akaike Information Criterion and the Baysian Information Criterion from the nlr4Results report.

In[465]:=
```
nlr4Results[{"BestFit", "BestFitParameters",
  "ParameterTable", "RSquared", "AdjustedRSquared", "AIC", "BIC"}]
```

Out[465]=
$$\left\{1000.\, e^{2.18151+0.4\left(0.\, t+5.\, \text{Log}\left[1.+e^{-4.71871\times 10^{-14}-0.2\, t}\right]-5.\, \text{Log}\left[1.+e^{1.6-0.2\, t}\right]\right)},\ \{\text{pAsym} \to 0.4,\ \text{k} \to 0.2,\ \text{tcd} \to -2.35935\times 10^{-13},\ \text{tcm} \to 8.\},\right.$$

	Estimate	Standard Error	t Statistic	P-Value
pAsym	0.4	2.13866×10^{-14}	1.87033×10^{13}	3.29869×10^{-91}
k	0.2	5.68276×10^{-16}	3.51942×10^{14}	3.94892×10^{-100}
tcd	-2.35935×10^{-13}	2.6039×10^{-13}	−0.906084	0.395009
tcm	8.	1.90777×10^{-13}	4.19338×10^{13}	1.15832×10^{-93}

$$\left., 1., 1., -555.336, -553.346\right\}$$

Extract the fitted values of the 4 parameters from the nlr4Results report and assign them to the 4 parameter variables.

In[466]:=
```
nlr4Results["BestFitParameters"]
```

Out[466]= $\{\text{pAsym} \to 0.4,\ \text{k} \to 0.2,\ \text{tcd} \to -2.35935\times 10^{-13},\ \text{tcm} \to 8.\}$

In[467]:=
```
pAsym = nlr4Results["BestFitParameters"][[1, 2]]
```

Out[467]= 0.4

In[468]:=
```
k = nlr4Results["BestFitParameters"][[2, 2]]
```

Out[468]= 0.2

In[469]:=
```
tcd = nlr4Results["BestFitParameters"][[3, 2]]
```

Out[469]= -2.35935×10^{-13}

In[470]:=
```
tcm = nlr4Results["BestFitParameters"][[4, 2]]
```

Out[470]= 8.

plot4GenExact is a ListPlot of the nCells4Pts exact data generated using the 4-parameter nCells4[t] function.

In[471]:=
```
plot4GenExact =
 ListPlot[nCells4Pts, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMax}},
  Frame → True, Axes -> False, PlotStyle → {PointSize → Medium, Blue},
  PlotLabel → ToString[nPts4] <> " nCells4Pts data at dt=" <> ToString[dt] <> " from " <>
    ToString[nParams] <> "-par. model", FrameLabel → {"time", "nCells4Pts data", "", ""}]
```

11 nCells4Pts data at dt=4. from 4−par. model

Out[471]=



plot4FitExact is a Plot of the nCells4[t] function using the currently assigned values of the 4 parameters fitted to the nCells4Pts exact data.

In[472]:=
```
plot4FitExact =
 Plot[nCells4[t], {t, tSt, tEn}, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMax}},
  Frame → True, Axes -> False, PlotStyle → {Blue},
  PlotLabel → ToString[nParams] <> "-par. nCells4[t] curve fitted to nCells4Pts data",
  FrameLabel → {"time", "nCells4[t]", "", ""}]
```

4−par. nCells4[t] curve fitted to nCells4Pts data

Out[472]=



Show the fitted 4-parameter nCells4[t] function curve and the nCells4Pts exact data values together on the same plot.

In[473]:=
```
plot4FitGenExact = Show[plot4FitExact, plot4GenExact,
  PlotLabel → ToString[nPts4] <> " nCells4Pts data at dt=" <>
    ToString[dt] <> " fitted with " <> ToString[nParams] <> "-par. nCells4[t]",
  FrameLabel → {"time", "nCells4Pts data & nCells4[t]", "", ""}]
```

11 nCells4Pts data at dt=4. fitted with 4−par. nCells4[t]

Out[473]=



Plot the pDiv4[t] and pMort4[t] functions together on the same plot using the currently assigned values of the 4 parameters fitted to the nCells4Pts exact data.

In[474]:=
```
plot4pDivpMortExact = Plot[{pDiv4[t], pMort4[t]}, {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True,
  PlotLabel → ToString[nParams] <> "-par. nCells4Pts fit pDiv4[t] (blue) & pMort4[t] (red)",
  FrameLabel → {"time", "pDiv4[t] & pMort4[t]"}]
```

4−par. nCells4Pts fit pDiv4[t] (blue) & pMort4[t] (red)

Out[474]=



Generate, fit and plot some noisy data (nCells4PtsRand) using the 4-parameter nCells4[t] function.

WARNING: Executing the following cell makes *Mathematica* always generate the same repeatable sequence of pseudo-random values from the specified seed value. Run without executing the next cell to restore pseudo-randomness or change the assigned seed value to get a different repeatable pseudo-random sequence.

In[475]:=
```
seed = 0; SeedRandom[seed]
```

maxNoise is the maximum allowed noise value. It must be less than or equal to the initial number of cells, n0.

In[476]:=
```
maxNoise = 200.;
```

nCells4PtsRand is the nCells4Pts list of ordered data pairs in which random noise has been added to the nCells4[t] coordinate values.

If you wish to fit and plot experimental data, assign them to nCells4PtsRand in the same form as shown below.

In[477]:=
```
nCells4PtsRand = Table[{t, nCells4[t] + RandomReal[{-maxNoise, maxNoise}]}, {t, tSt, tEn, dt}]
```

Out[477]=
```
{{0., 1060.99}, {4., 1841.96}, {8., 3272.69}, {12., 5044.27}, {16., 6817.4}, {20., 7912.95},
 {24., 8209.81}, {28., 8662.14}, {32., 8585.03}, {36., 8865.83}, {40., 8700.01}}
```

nPts4Rand is the number of generated data points in the nCells4PtsRand list.

In[478]:=
```
nPts4Rand = Length[nCells4PtsRand]
```

Out[478]=
```
11
```

Assign the maximum y-axis value to be used in the noisy data plots to yMaxRand.

In[479]:=
```
Max[nCells4PtsRand[[All, 2]]]
```

Out[479]=
```
8865.83
```

In[480]:=
```
yMaxRand = 9000.;
```

Clear all the variables to be used by NonlinearModelFit.

In[481]:=
```
Clear[nlr4ResultsRand, pAsym, k, tcd, tcm]
```

Note that since the data are dependent on random numbers, the nonlinear fit may be very sensitive to the initial guesses of the values of the 4 parameters.

Call the NonlinearModelFit function which adjusts the values of the 4 parameters to give the best fit of the nCells4[t] function to the nCells4PtsRand noisy data and return a report in nlr4ResultsRand. All 4 parameters are unconstrained.

In[482]:=
```
nlr4ResultsRand = NonlinearModelFit[nCells4PtsRand, nCells4[t],
   {{pAsym, 0.5}, {k, 0.5}, {tcd, 1.}, {tcm, 10.}}, t, MaxIterations → 500]
```

Out[482]=
```
FittedModel[ 1000. e^(2.16844+≪1≫) ]
```

Show the best fitted model, the best fit parameters, the parameter table, the R-squared coefficient of determination, the value of R-squared adjusted for the number of model parameters, the Akaike Information Criterion and the Baysian Information Criterion from the nlr4ResultsRand report.

In[483]:=
```
nlr4ResultsRand[{"BestFit", "BestFitParameters",
  "ParameterTable", "RSquared", "AdjustedRSquared", "AIC", "BIC"}]
```

Out[483]=
$$\left\{1000.\, e^{2.16844+0.1906\,(0.\,t+4.22826\,\text{Log}\left[1.+e^{-1.61937-0.236504\,t}\right]-4.22826\,\text{Log}\left[1.+e^{2.81308-0.236504\,t}\right])},\right.$$

$$\{\text{pAsym} \to 0.1906,\ \text{k} \to 0.236504,\ \text{tcd} \to -6.84712,\ \text{tcm} \to 11.8944\},$$

	Estimate	Standard Error	t Statistic	P-Value
pAsym	0.1906	0.0824293	2.31228	0.0540048
k	0.236504	0.0315367	7.4993	0.000137384
tcd	−6.84712	12.6427	−0.541589	0.604903
tcm	11.8944	2.93081	4.05841	0.00481924

$$\left., 0.999855, 0.999771, 138.413, 140.403\right\}$$

Extract the fitted values of the 4 parameters from the nlr4ResultsRand report and assign them to the 4 parameter variables.

In[484]:=
```
nlr4ResultsRand["BestFitParameters"]
```

Out[484]=
$$\{\text{pAsym} \to 0.1906,\ \text{k} \to 0.236504,\ \text{tcd} \to -6.84712,\ \text{tcm} \to 11.8944\}$$

In[485]:=
```
pAsym = nlr4ResultsRand["BestFitParameters"][[1, 2]]
```

Out[485]= 0.1906

In[486]:=
```
k = nlr4ResultsRand["BestFitParameters"][[2, 2]]
```

Out[486]= 0.236504

In[487]:=
```
tcd = nlr4ResultsRand["BestFitParameters"][[3, 2]]
```

Out[487]= −6.84712

In[488]:=
```
tcm = nlr4ResultsRand["BestFitParameters"][[4, 2]]
```

Out[488]= 11.8944

plot4GenRand is a ListPlot of the nCells4PtsRand noisy data generated using the 4-parameter nCells4[t] function.

In[489]:=
```
plot4GenRand =
 ListPlot[nCells4PtsRand, PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMaxRand}},
  Frame → True, Axes -> False, PlotStyle → {PointSize → Medium, Red},
  PlotLabel → ToString[nPts4Rand] <> " nCells4PtsRand data at dt=" <>
    ToString[dt] <> " from " <> ToString[nParams] <> "-par. model",
  FrameLabel → {"time", "nCells4PtsRand data", "", ""}]
```

11 nCells4PtsRand data at dt=4. from 4−par. model

Out[489]=



plot4FitRand is a Plot of the nCells4[t] function using the currently assigned values of the 4 parameters fitted to the nCells4PtsRand noisy data.

In[490]:=
```
plot4FitRand = Plot[nCells4[t], {t, tSt, tEn},
  PlotRange → {{tSt - minOffset, tEn}, {0. - minOffset, yMaxRand}},
  Frame → True, Axes -> False, PlotStyle → {Red},
  PlotLabel → ToString[nParams] <> "-par. nCells4[t] curve fitted to nCells4PtsRand data",
  FrameLabel → {"time", "nCells4[t]", "", ""}]
```

4−par. nCells4[t] curve fitted to nCells4PtsRand data

Out[490]=



Show the fitted 4-parameter nCells4[t] function curve and the nCells4PtsRand noisy data values together on the same plot.

In[491]:=
```
plot4FitGenRand = Show[plot4FitRand, plot4GenRand,
  PlotLabel → ToString[nPts4Rand] <> " nCells4PtsRand data at dt=" <>
    ToString[dt] <> " fitted with " <> ToString[nParams] <> "-par. nCells4[t]",
  FrameLabel → {"time", "nCells4PtsRand data & nCells4[t]", "", ""}]
```

11 nCells4PtsRand data at dt=4. fitted with 4−par. nCells4[t]

Out[491]=



Plot the pDiv4[t] and pMort4[t] functions together on the same plot using the currently assigned values of the 4 parameters fitted to the nCells4PtsRand noisy data.

In[492]:=
```
plot4pDivpMortRand = Plot[{pDiv4[t], pMort4[t]}, {t, tSt, tEn}, PlotRange →
   {{tSt - minOffset, tEn}, {0. - minOffset, yDivMortMax + minOffset}}, Frame → True, PlotLabel →
   ToString[nParams] <> "-par. nCells4PtsRand fit pDiv4[t] (blue) & pMort4[t] (red)",
  FrameLabel → {"time", "pDiv4[t] & pMort4[t]"}]
```

4−par. nCells4PtsRand fit pDiv4[t] (blue) & pMort4[t] (red)

Out[492]=